
Chapter 7
Selection – If and Select
This chapter explains:
· how to use If and Select statements to carry out tests

· how to use operators such as >
· how to use And, Or and Not
· how to declare and use Boolean data
l Introduction

We all make decisions in daily life. We wear a coat if it is raining. We buy a CD if we have enough money. Decisions are also used a lot in programs. The computer tests a value and according to the result, takes one course of action or another. Whenever the program has a choice of actions and decides to take one action or the other, an If or a Select statement is used to describe the situation.

We have seen that a computer program is a series of instructions to a computer. The computer obeys the instructions one after another in sequence. But sometimes we want the computer to carry out a test on some data and then take one of a choice of actions depending on the result of the test. For example, we might want the computer to test someone’s age and then tell them either that they may vote or that they are too young. This is called selection. It uses a statement (or instruction) called the If state​ment, the central subject of this chapter.

If statements are so important that they are used in every programming language that has ever been invented.

l The If statement

Our first example is a program that simulates the digital lock on a safe. The screen is as shown in figure 7.1. The safe is locked unless the user enters the correct code into a text box. The text box is initially emptied when the form is designed. The program compares the text that is entered with the correct code. If the code is correct, a message is displayed.

Private Sub Button1_Click(ByVal sender As System.Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    Dim code As String

    Label2.Text = ""

    code = TextBox1.Text

    If code = "bill" Then

        Label2.Text = "unlocked"

    End If

End Sub

Figure 7.1 Screen for the safe program.

The If statement tests the value of the string. If the string equals the value "bill", the statement sandwiched between the If statement and the End If statement is carried out. Next any statement after the End If is executed. On the other hand, if the string is not equal to "bill", the sandwiched statement is ignored and any statement after the End If is executed.

One way of visualizing an If statement is as an activity diagram (Figure 7.2). This shows the above If statement in graphical form. To use this diagram, start at the blob at the top and follow the arrows. A decision is shown as a diamond, with the two possible conditions shown in square brackets. Actions are shown in rounded boxes and the end of the sequence is a specially-shaped blob at the bottom of the diagram.

Figure 7.2 Activity diagram for an If statement.
There are two parts to the If statement:
· the condition being tested;

· the statement or sequence of statements to be executed if the condition is true.

All programs consist of a sequence of actions, and the sequence evident here is:

1.
A piece of text is input from the text box.

2.
Next a test is done.

3.
If appropriate, a message is displayed to say that the safe is unlocked.

Very often we want not just one, but a complete sequence of actions carried out if the result of the test is true, and these are sandwiched between the If statement and the End If statement. 
Indentation

Notice that the lines are indented to reflect the structure of this piece of program. (Indentation means using spaces to push the text over to the right.) The VB development system does this automatically when you type in an If statement. Although indentation is not essential , it is highly desirable so that the (human) reader of a program can understand it easily. All good programs (whatever the language) use indentation and all good programmers use it.

If...Else

Sometimes we want to specify two sequences of actions – those that are carried out if the condition is true and those that are carried out if the condition is false.

The user of the voting checker program enters their age into a text box and the program decides whether they can vote or not. The screen is shown in Figure 7.3. When the user clicks on the button, the program extracts the information that the user has entered into the text box, converts the string into an integer and places the number in the variable called age. Next we want the program to take different actions depending on whether the value is:

· greater than 17

or

· less than or equal to 17.

Then the results of the test are displayed in a number of labels.

Figure 7.3 The voting checking program screen

Private Sub Button1_Click(ByVal sender As System.Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    Dim age As Integer

    age = CInt(TextBox1.Text)

    If age > 17 Then

        DecisionLabel.Text = "you may vote"

        CommentaryLabel.Text = "congratulations"

    Else

        DecisionLabel.Text = "you may not vote"

        CommentaryLabel.Text = "sorry"

    End If

    SignOffLabel.Text = "Best Wishes"

End Sub
There are three parts to this If statement:
· the condition being tested – in this case whether the age is greater than 17

· the statement or sequence of statements to be executed if the condition is true;

· the statement or statements to be executed if the condition is false.

The new element here is the word Else, which introduces the second part of the If statement. Notice again how the indentation helps to emphasize the intention of the program. 

We can visualize an If...Else statement as an activity diagram, as shown in Figure 7.4. The diagram shows the condition being testing and the two separate actions.

Figure 7.4 Activity diagram for an If...Else statement.
l Comparison operators

The programs above used some of the comparison operators. Here is a complete list:

	symbol
	means

	>
	 greater than

	< 
	 less than

	=
	 equals

	<>
	 not equal to

	<= 
	 less than or equal to

	>= 
	 greater than or equal to


Notice that VB uses the equals sign (=) to test whether two things are equal. This same symbol is also used to assign a value to a variable. 

Choosing the appropriate operator often has to be done with great care. In the program to test whether someone can vote, the appropriate test should probably be:

If age >= 18 Then

    Label2.Text = "you can vote"

End If

Note that it is usually possible to write conditions in either of two ways. The following two program fragments achieve exactly the same result, but use different conditions:

If age >= 18 Then

    Label2.Text = "you may vote"

Else

    Label2.Text = "sorry"

End If

achieves the same end as:

If age < 18 Then

    Label2.Text = "sorry"

Else

    Label2.Text = "you may vote"

End If

Although these two fragments achieve the same end result, the first is probably better, because it spells out more clearly the condition for eligibility to vote.

SELF-TEST QUESTION
7.1 Do these two pieces of VB achieve the same end or not?
If age > 18 Then

    Label2.Text = "you may vote"

End If

If age < 18 Then

    Label2.Text = "you may not vote"

End If

l And, Or, Not

Often in programming we need to test two things at once. Suppose, for example, we want to test whether someone should pay a junior rate for a ticket:

If age > 6 And age < 16 Then

    Label1.Text = "junior rate"

End If

The word And is one of the VB logical operators and simply means ‘and’ as we would use it normal language.

Brackets can be used to improve the readability of these more complex conditions. For example we can re-write the above statement as:

If (age > 6) And (age < 16) Then

    Label1.Text = "junior rate"

End If

Although the brackets are not essential, they serve to distinguish the two conditions being tested.

It might be very tempting to write:

If age > 6 And < 18 Then    ' error!

but this is incorrect because the conditions have to be spelled out in full as follows:

If age > 6 And age < 18
Then    ' OK

We would use the Or operator in an If statement like this:

If age < 6 Or Age > 60 Then

    Label1.Text = "reduced rate"

End If

in which the reduced rate is applicable for people who are younger than 6 or older than 60.

The Not operator, with the same meaning as in English, gets a lot of use in programming, even though in English the use of a negative can suffer from lack of clarity. Here is an example of the use of not:
If Not (age > 18) Then

    Label1.Text = "too young"

End If

This means: test to see if the age is greater than 18. If this result is true, make it false. If it is false, make it true. Then, if the outcome is true, display the message. This can, of course, be written more simply without the Not operator.

SELF-TEST QUESTION
7.2
Rewrite the above If statement without using the Not operator.
This next program illustrates a more complex series of tests. Two dice are thrown in a betting game and the program has to decide what the result is. We will create two track bars, each with a range of 1 to 6 to specify the values of each of the two die (Figure 7.5). To start with, we make the rule that only a total score of six wins anything.

Figure 7.5 The dice program, version 1.

The program code is given below. Whenever either of the two track bars is moved , the method is called to display the total value and decide whether a win has occurred.

Private Sub TrackBar1_Scroll(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles TrackBar1.Scroll
    CheckValues()

End Sub

Private Sub TrackBar2_Scroll(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles TrackBar2.Scroll
    CheckValues()

End Sub

Private Sub CheckValues()

    Dim die1, die2, total As Integer

    die1 = TrackBar1.Value

    die2 = TrackBar2.Value

    total = die1 + die2

    Label1.Text = "total is " & total

    If total = 6 Then

      Label2.Text = "you have won"

    Else

      Label2.Text = "you have lost"

    End If

End Sub
Now we will alter the rules and see how to rewrite the program. Suppose that any pair of identical values wins, i.e. two ones, two twos etc. Then the If statement is:

If die1 = die2 Then

    Label2.Text = "you have won"
End If

Now let’s suppose that you only win if you get a total of either 2 or 7:
If  (total = 2) Or (total = 7) Then

    Label2.Text = "you have won"

End If

Notice that we have enclosed each of the conditions with brackets. These brackets aren’t strictly necessary in VB, but they help a lot to clarify the meaning of the condition to be tested.

SELF-TEST QUESTION
7.3
Alter the program so that a win is a total value of 2, 5 or 7.
SELF-TEST QUESTION
7.4
Write If statements to test whether someone is eligible for full-time employment. The rule is that you must be 16 or above and younger than 65.
l Nested Ifs and ElseIf
Look at the following program fragment:

If age > 6 Then

    If age < 16 Then

        Label1.Text = "junior rate"

    Else

        Label1.Text = "adult rate"

    End If

Else

    Label1.Text = "child rate"

End If

You will see that the second If statement is completely contained within the first. (The indentation helps to make this clear.) This is called nesting. Nesting is not the same as indentation – it is just that the indentation makes the nesting very apparent. The meaning of this nested code is as follows:

· If the age is greater than 6, then the second If is carried out.

· If the age is not greater than 6, then the Else part is carried out
The overall effect of this piece of program is:

· If the age is greater than 6 and less than 16, the rate is the junior rate.

· If the age is greater than 6 but not less than 16, the rate is the adult rate.

· If the age is not greater than 6, the rate is the child rate.

It is common to see nesting in programs, but a program like this has a complexity which makes it slightly difficult to understand. Often it is possible to write a program more simply using the logical operators and/or a variation of the If statement that uses ElseIf. Here, for example, the same result as above is achieved without nesting:

If (age > 6) And (age < 16) Then

    Label1.Text = "junior rate"

ElseIf age >= 16 Then

    Label1.Text = "adult rate"

Else

    Label1.Text = "child rate"

End If

The If...ElseIf statement describes a series of mutually exclusive choices. The first condition follows the initial If. Subsequent conditions follow any number of ElseIf keywords. Optionally there is a final Else to address any condition that has not already been met. There is a single End If at the end of the complete statement.

We now have two pieces of program that achieve the same end result, one with nesting and one without. Some people argue that it is hard to understand nesting, such a program is prone to errors and that therefore nesting should be avoided. Nesting can always be avoided using either logical operators or ElseIf or both.

SELF-TEST QUESTION
7.5
Write a program to input a salary from a track bar and determine how much tax someone should pay according to the following rules:


  People pay no tax if they earn up to $10,000. They pay tax at the rate of 20% on the amount they earn over $10,000 but up to $50,000. They pay tax at 90%on any money they earn over $50,000. The track bar should have a range from 0 to 100,000.
In the next program we create two track bars, one called Tom, the other called Jerry. The program compares the values and reports on which one is set to the larger value. The screen is shown in Figure 7.6. The library method FillRectangle is used to draw a solid rectangle whose width across the screen is equal to the value obtained from the corresponding track bar.

Private Sub TrackBar1_Scroll(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles TrackBar1.Scroll
    CompareValues()

End Sub

Private Sub TrackBar2_Scroll(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles TrackBar2.Scroll
    CompareValues()

End Sub

Private Sub CompareValues()

    Dim paper As Graphics

    paper = PictureBox1.CreateGraphics()

    Dim myBrush As New SolidBrush(Color.Black)

    Dim tomValue, jerryValue As Integer

    tomValue = TrackBar1.Value

    jerryValue = TrackBar2.Value

    paper.Clear(Color.White)

    paper.FillRectangle(myBrush, 10, 10, tomValue, 20)

    paper.FillRectangle(myBrush, 10, 50, jerryValue, 20)

    If tomValue > jerryValue Then

        Label1.Text = "Tom is bigger"

    Else

        Label1.Text = "Jerry is bigger"

    End If

End Sub

Figure 7.6 Screen for the Tom and Jerry program.

This program works fine, but again illustrates the importance of care when you use If statements. In this program, what happens when the two values are equal? The answer is that the program finds that Jerry is bigger – which is clearly not the case. We could enhance the program to spell things out more clearly by changing the If statement to:
If jerryValue > tomValue Then

    Label1.Text = "Jerry is bigger"

ElseIf tomValue > jerryValue Then

    Label1.Text = "Tom is bigger"

Else

    Label1.Text = "They are equal"

End If

This is another illustration of an If statement with an ElseIf part. 

SELF-TEST QUESTION
7.6
Write a program that creates three track bars and displays the largest of the three values.
This next example is a program that keeps track of the largest value of a number as it changes. Some stereo amplifiers have a display that shows the volume being created. The display waxes and wanes according to the volume at any point in time. Sometimes the display has an indicator that shows the maximum value that is currently being output. This program displays the numerical value of the maximum value that the track bar is set to (see Figure 7.7). It uses a single If statement that compares the current value of the track bar with the value of a variable named max, a class level variable that holds the value of the largest value achieved so far. max is declared like this:

Private max As Integer = 0

and the method to handle track bar events is:

Figure 7.7 Screen for the amplifier display.

Private Sub TrackBar1_Scroll(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles TrackBar1.Scroll
    Dim volume As Integer

    volume = TrackBar1.Value

    If volume > max Then

        max = volume

    End If

    Label1.Text = "maximum value is " & CStr(max)

End Sub

SELF-TEST QUESTIONS
7.7
Write a program that displays the numerical value of the minimum value that the track bar is set to.

7.8
The Young and Beautiful holiday company restricts its clients to ages between 18 and 30. (Below 18 you have no money; after 30 you have too many wrinkles.) Write a program to test whether you are eligible to go on holiday with this company.
We now return to the dice-throwing program discussed earlier. Instead of inputting the dice values via the track bars, we change the program so that the computer decides the die values randomly. We will create a button, labeled ‘Throw’. When it is clicked, the program will obtain two random numbers and use them as the die values (Figure 7.8).

Figure 7.8 The dice program, version 2.

To get a random number in VB, we create an object from the library class Random and then use its method Next. This method returns a random number, an Integer in any range we choose, specified by the parameters. We met this class back in chapter 6.

The program to throw two dice is given below. At class level we declare:
Private randomNumber As Random = New Random()

and then the event handling method is:

Private Sub Button1_Click(ByVal sender As Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click
    Dim die1, die2 As Integer

    die1 = randomNumber.Next(1, 6)

    die2 = randomNumber.Next(1, 6)

    Label1.Text = "the die values are " _

                 & CStr(die1) & " and " & CStr(die2)

    If die1 = die2 Then

        Label2.Text = "dice equal - a win"

    Else

        Label2.Text = "dice not equal - lose"

    End If

End Sub
l Select
The Select statement is another way of doing a lot of If statements. You can always accomplish everything you need with the aid of If statements but Select can be useful in appropriate circumstances. For example, suppose we need a piece of program to display the day of the week as a string. Suppose that the program represents the day of the week as an Integer variable called dayNumber, which has one of the values 1 to 7, representing the days Monday to Sunday. We want to convert the integer version of the day into a string version called dayName. We could write the following series of If statements:

If dayNumber = 1 Then

    dayName = "Monday"

ElseIf dayNumber = 2 Then

    dayName = "Tuesday"

ElseIf dayNumber = 3 Then

    dayName = "Wednesday"

ElseIf dayNumber = 4 Then

    dayName = "Thursday"

ElseIf dayNumber = 5 Then

    dayName = "Friday"

ElseIf dayNumber = 6 Then

    dayName = "Saturday"

ElseIf dayNumber = 7 Then

    dayName = "Sunday"

End If

Now although this piece of coding is neat, clear and well-structured, there is an alternative that has the same effect using the Select statement:

Select Case dayNumber

    Case 1 

        dayName = "Monday"

    Case 2 

        dayName = "Tuesday"

    Case 3 

        dayName = "Wednesday"

    Case 4 

        dayName = "Thursday"

    Case 5 

        dayName = "Friday"

    Case 6 

        dayName = "Saturday"

    Case 7 

        dayname = "Sunday"

End Select
This now exploits the symmetry of what needs to happen more clearly than the equivalent series of ElseIfs. Notice that the complete Select statement ends with a End Select statement.

A Select statement like this can be visualized as an activity diagram in Figure 7.9.

Figure 7.9 Activity diagram showing part of a Select statement.

SELF-TEST QUESTION
7.9
Write a method that converts the integers 1, 2, 3 and 4 into the words diamonds, hearts, clubs and spades respectively.
Several statements can follow one of the options in a Select statement. For example, one of the options could be:

Case 6

    MessageBox.Show("hurray")

    dayName = "Saturday"

Another feature of the Select statement is grouping several options together, like this:

Select Case dayNumber 

Case 1, 2, 3, 4, 5

    dayName = "weekday"

Case 6, 7

    dayName = "weekend"

End Select

Another, sometimes useful, part of the Select statement is the Else option. Suppose in the above example that the value of the integer denoting the day of the week is input from a text box. Then there is the distinct possibility that the user will erroneously enter a number that is not in the range 0 to 7. Any decent program needs to take account of this, in order to prevent something odd happening or the program crashing. The Select statement is very good at dealing with this situation, because we can supply a ‘catch-all’ or default option that will be used if none of the others are valid:

Select Case dayNumber

    Case 1 

        dayName = "Monday"

    Case 2 

        dayName = "Tuesday"

    Case 3 

        dayName = "Wednesday"

    Case 4 

        dayName = "Thursday"

    Case 5 

        dayName = "Friday"

    Case 6 

        dayName = "Saturday"

    Case 7 

        dayName = "Sunday"

    Case Else 

        dayName = "illegal day"

End Select

If an Else option is not written as part of a Select statement and if none of the cases provided corresponds to the actual value of the variable, then all the options are ignored.

The Select statement is very useful, but unfortunately it is not as flexible as it could be. Suppose, for example, we want to write a piece of program to display two numbers, with the smaller first, followed by the larger. Using If statements, we would write:

If a > b Then

    Label1.Text = CStr(a) & " is greater than " & CStr(b)

ElseIf b > a Then

    Label1.Text = CStr(b) & " is greater than " & CStr(a)

Else

    Label1.Text = "they are equal"

End If

We may be tempted to rewrite this using a Select statement as follows:
Select Case ? ' beware! illegal VB

    Case a > b

        Label1.Text = CStr(a) & " is greater than" & CStr(b)
    Case b > a

        Label1.Text = CStr(b) & " is greater than" & CStr(a)
    Case a = b

        Label1.Text = "they are equal"

End Select

but this is not allowed because, as indicated by the question mark, Select only works with a single integer or string variable as its subject.

l Boolean variables

All of the types of variable that we have met so far are designed to hold numbers or strings. Now we meet a new kind of variable called a Boolean, that can only hold either the value True or the value False. The words Boolean, True and False are reserved keywords in VB and cannot be used for any other purpose. This type of variable is named after the 19th century British mathematician George Boole who made a large contribution towards the development of mathematical logic, in which the ideas of true and false play a central role.

We will introduce Boolean variables by looking at the properties of the labels that are available from the toolbox. Figure 7.10 shows a display that can either be switched on or switched off using buttons. When the user clicks on the buttons the Visible property of the label is changed:

Private Sub OffButton_Click(ByVal sender As Object, _

                            ByVal e As System.EventArgs) _

                            Handles OffButton.Click

    Label1.Visible = False

End Sub

Private Sub OnButton_Click(ByVal sender As Object, _

                           ByVal e As System.EventArgs) _

                           Handles OnButton.Click

    Label1.Visible = True

End Sub

 Figure 7.10 The fast food sign

The Visible property is a Boolean and we can change the value as shown above. We can also test its value using If statements. So we could re-write the program so that it has a single button that switches the display from visible to invisible, or vice versa, using the statements:

Private Sub Button1_Click(ByVal sender As Object, _

                          ByVal e As System.EventArgs) _

                          Handles Button1.Click

    If Label1.Visible = True Then

        Label1.Visible = False

    Else

        Label1.Visible = True

    End If

End Sub
Now that we have seen how to use a Boolean property, we now look at declaring  our own Boolean variables. We can declare a variable of type Boolean like this:
Dim finished As Boolean

and we can assign either of the values True and False:

finished = True

Equally importantly, we can test the value of a Boolean in an If statement:

If finished Then

if (MessageBox.Show("Good bye")

End If

The value of finished is tested, and if it is True the accompanying statement is executed. An equivalent, but slightly more cumbersome, way of writing the same test is:

If finished = True Then

    MessageBox.Show("Good bye")

End If

Boolean variables are used in programming to remember something, perhaps for a short time, perhaps for the whole time that the program is running. As an example, we look at a program, figure 7.11, that draws a rectangle in a picture box when a button is clicked. Sometimes we want the rectangle to be drawn in outline and sometimes we want it drawn filled in. We provide two buttons allowing the user to specify their option. Once a button has been clicked, the program remembers the option until the user clicks again on a button.

Figure 7.11 The display from the remember program.

The class level declaration of the Boolean variable is:

Private filled As Boolean = True

This is the variable that remembers what the user has specified. It either has the value True (to denote that rectangles should be filled in) or it has the value False.

The methods to handle button clicks simply make the Boolean True or False as appropriate:

Private Sub FillButton_Click(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles FillButton.Click

    filled = True

End Sub

Private Sub UnFillButton_Click(ByVal sender As Object, _

                               ByVal e As System.EventArgs) _

                               Handles UnFillButton.Click

    filled = False

End Sub

When the user clicks on the draw button, the program  first tests the value of the Boolean value filled using an If statement and then either draws a filled or an unfilled rectangle.

Private Sub DrawButton_Click(ByVal sender As Object, _

                             ByVal e As System.EventArgs) _

                             Handles DrawButton.Click

    Dim paper As Graphics = PictureBox1.CreateGraphics()

    Dim myPen As Pen = New Pen(Color.Black)

    Dim myBrush As SolidBrush = New SolidBrush(Color.Black)

    paper.Clear(Color.White)

    If filled = True Then

        paper.FillRectangle(myBrush, 10, 10, 50, 50)

    Else

        paper.DrawRectangle(myPen, 10, 10, 50, 50)

    End If

End Sub

Methods can use Boolean values as parameters and as return values. For example, here is a method that checks whether three numbers are in numerical order:
Private Function inOrder(ByVal a as Integer, _

                         ByVal b As Integer, _

                         ByVal c As Integer) As Boolean

    If (a <= b) And (b <= c) Then

        Return True

    Else

        Return False

    End If

End Sub

l Programming principles

The computer normally obeys instructions one-by-one in a sequence. An If statement instructs the computer to test the value of some data and then take one of a choice of actions depending on the result of the test. This choice is sometimes called selection. The test of the data is called a condition. After an If statement is completed, the computer continues obeying the instructions in sequence.

l Programming pitfalls


You might find that you have written an If statement like this:

If a > 18 And < 25 Then


which is wrong. Instead, the And must link two complete conditions,


preferably in brackets for clarity, like this:


If (a > 18) And (a < 25) Then

l Grammar spot

The first kind of If statement has the structure:
If condition Then

    statements

End If

The second type of If statement has the structure:
If condition Then

    statements

Else

    statements

End If

The third type of If statement has the structure that follows this pattern:

If condition Then

    statements

ElseIf condition Then

    statements

ElseIf condition Then

    statements

Else

    statements

End If

The final Else is optional.

The Select statement has the structure:

Select Case variable

    Case value1 

        statements

    Case value2 

        statements

    Case Else 

        statements

End Select

The final Case Else is optional.

l New language elements

· Control structures for decisions:


If, Then, ElseIf, Else, End If


Select, Case, Else, End Select

· The comparison operators >, <, =, <>, <= and >=

· The logical operators And Or Not

· Variables declared as Boolean, which can take either the value True or the value False.

SUMMARY

l
If statements allow the programmer to control the sequence of actions by making the program carry out a test. Following the test, the computer carries out one of a choice of actions.

l
There are three varieties of If statement:

If...Then...End If

If...Then...Else...End If

If...Then...ElseIf...End If

l
A Boolean variable can be assigned the value True or the value False. A Boolean variable can be tested with an If statement.

l
The Select statement provides a convenient way of carrying out a number of tests. However, the Select statement is restricted to tests on integers or on strings.

EXERCISES

7.1 Deal a card Write a program with a single button on it which, when clicked, randomly selects a single playing card. First use the random number generator in the library to create a number in the range 1 to 4. Then convert the number to a suit (heart, diamond, club, spade). Next use the random number generator to create a random number in the range 1 to 13. Convert the number to an ace, 2, 3 etc. and finally display the value of the chosen card. (Hint: use Select as appropriate.)

7.2 Sorting Write a program to input numbers from three track bars, or three text boxes, and display them in increasing numerical size.

7.3 Cinema (movie theatre) price Write a program to work out how much a person pays to go to the cinema. The program should input an age from a track bar or a text box and then decide on the following basis:

· under 5, free;

· aged 5 to 12, half price;

· aged 13 to 54, full price;

· aged 55, or over, free.

7.4 Betting A group of people are betting on the outcome of three throws of the dice. A person bets $1 on predicting the outcome of the three throws. Write a program that uses the random number method to simulate three throws of a die and displays the winnings according to the following rules:

· all three throws are sixes: win $20;

· all three throws are the same (but not sixes): win $10;

· any two of the three throws are the same: win $5.

7.5 Digital combination safe Write a program to act as the digital combination lock for a safe. Create three buttons, representing the numbers 1, 2 and 3. The user clicks on the buttons, attempting to guess the correct numbers (say 331121). The program remains unhelpfully quiet until the correct buttons are pressed. Then it congratulates the user with a suitable message. A button is provided to allow users to give up when they become frustrated.

Enhance the program so that it has another button which allows the user to change the safe‘s combination.

7.6 Rock, scissors, paper game This is a game for two humans to play. Each player simultaneously chooses one of rock, scissors or paper. Rock beats scissors, paper beats rock and scissors beats paper. If both players choose the same, it is a draw. Write a program to play the game. The player selects one of three buttons, marked rock, scissors or paper. The computer makes its choice randomly using the random number generator. The computer also decides and displays who has won.

7.7 The calculator Write a program which simulates a simple desk calculator (Figure 7.12) that acts on integer numbers. It has one button for each of the 10 digits, 0 to 9. It has a button to add and a button to subtract. It has a clear button, to clear the display, and an equals (=) button to get the answer.

Figure 7.12 The calculator.

When the clear button is pressed the display is set to zero and the (hidden) total is set to zero.

When a digit button is pressed, the digit is added to the right of those already in the display (if any).

When the + button is pressed, the number in the display is added to the total (and similarly for the – button). The display is set equal to zero.

When the = button is pressed, the value of the total is displayed.

7.8 The elevator Write a program to simulate a very primitive elevator. The elevator is represented as a filled rectangle. There are two buttons – one to make it move up the screen and one to make it move down.

7.9 Nim is a game played with matchsticks (unused or used, it does not matter). It doesn‘t 

matter how many matches there are. The matches are put into three piles. Again, it doesn‘t matter how many matches there are in each pile. Each player goes in turn. A player can remove any number of matches from any one pile, but only one pile. A player must remove at least one match. The winner is the person who causes the other player to take the last match.

Write a program to play the game. Initially the computer deals three piles, with a random number (in the range 1 to 200) of matches in each pile. One player is the computer, which chooses a pile and an amount randomly. The other player is the human user, who specifies the pile number and quantity using text boxes, before clicking on a ‘go’ button.

7.10 Turtle graphics Turtle graphics is a way of making programming easy for young children. Imagine a pen fixed to the belly of a turtle. As the turtle crawls around a floor, the pen draws on the floor. The turtle can be issued with commands as follows:

·  pen up

·  pen down

·  turn left 90°

·  turn right 90°

·  go forward n pixels

Initially the turtle is at coordinates 0, 0 and facing to the right.

So, for example, we can draw a rectangle using the sequence:

1. pen down

2. go forward 20 pixels

3. turn right 90°

4. go forward 20 pixels

5. turn right 90°

6. go forward 20 pixels

7. turn right 90°

8. go forward 20 pixels

Write a program that behaves as the turtle, with one button for each of the commands. The number of pixels, n, to be moved is input via a track bar or a text box.

ANSWERS TO SELF-TEST QUESTIONS

7.1


No, because they treat the particular age of 18 differently. The two If statements also 


create different messages.


7.2


If age <= 18

    Label1.Text = "too young"

End If


7.3
If (total = 2) Or (total = 5) Or (total = 7) Then

    Label2.Text = "you have won"

End If

7.4


If age >= 16 And age < 65 Then

    MessageBox.Show("you are eligible")
End If


7.5


Dim salary, tax As Integer

salary = TrackBar1.Value

If (salary > 10000) And (salary <= 50000) Then

    tax = (salary - 10000)/5

ElseIf salary > 50000 Then

    tax = 8000 + ((salary - 50000) * 9 / 10)

Else

    tax = 0

End If

7.6


Dim a, b, c As Integer

a = TrackBar1.Value

b = TrackBar2.Value

c = TrackBar3.Value

If a > b And a > c Then

    largest = a

ElseIf b > a And b > c Then

    largest = b

ElseIf c > a And c > b Then

    largest = c

End If

MessageBox.Show("largest value is " & CStr(largest))

7.7


The essential part of this program is:
If volume < min Then

    min = volume

End If

Label2.text = "Minimum value is " & CStr(min)


7.8


Dim age As Integer

age = CInt(TextBox1.Text)

If age >= 18 And age <= 30 Then

  TextBox2.Text = "you are eligible"  

EndIf


7.9

Function Convert(ByVal s As Integer) As String

    Dim suit As String

    Select Case s

        Case 1  

            suit = "diamonds" 

        Case 2  

            suit = "hearts" 

        Case 3  

            suit = "clubs" 

        Case 4  

            suit = "spades" 

    End Select

    Return suit

End Function

